
Part 2 – Custom DOE Constraints 
 

Using JMP’s Custom DOE platform to create an appropriate experimental design 
JMP’s DOE Custom Designer has many features that are helpful in creating a customized experimental design.  In this 
case, we wanted to employ a mixture design.  However, for this experimental design we were dealing with a special 
requirement:  One of the mixture components could consist of one of eight different materials, each of which had its 
own range of fractions that it comprised in the mixture.  In addition, initially we wanted to study an additional 3 mixture 
factors in the experiment.   

The solution involved using “Disallowed Combinations Script” to define constraints, but also involved some special 
scripting.  The process is described below: 

1.  Define Factors 
a. The first factor describes the eight material types of the special mixture component.  This is a categorical 

factor which we will call Additive, which has eight levels (Additive 1, Additive 2, Additive 3, Additive 4, 
Additive 5, Additive 6, Additive 7, Additive 8.) 

b. The second factor (M1) is defined as a continuous variable, representing the fractional level of the 
material defined by Additive.  When defining this factor, we let the min and max range from 0 to 1.  
Individual levels for each level of Additive will be specified in Constraints.  

c. M2, M3, and M4 are the remaining three mixture components, all defined as Continuous, and each with 
its own range of possible mixture fractions. 

d. Other factors which are independent of the mixture can be added as needed, but are not shown here. 
2. Set Constraint for overall mixture: 

a. Mixture must sum to 100%.  One way to define this would be to disallow any combination of 
M1+M2+M3+M4 that does not add exactly to 1.0.  However, the algorithm that JMP uses to find an 
optimal design needs a little “flexibility” to operate efficiently.  We do this by specifying that 
M1+M2+M3+M4 needs to fall in the range of 0.99 to 1.01.  But since this is a “Disallowed Combinations” 
script, we want to define the constraint as the mixture sum cannot be less than 0.99, nor greater than 
1.01.  To do this we employ inequalities and an OR statement as follows: 

M1+M2+M3+M4 > 1.01  |  M1+M2+M3+M4 < 0.99 
3. Set constraint on M1: 

a. The range of M1 depends on the material selected by Additive type.  For material Additive 1, let’s say 
that M1 needs to fall between 0.5 and 0.6 of the mixture.  Since we are again dealing with “disallowed 
combinations”, we have to state what the range “can’t be.” 

Additive == “Additive 1” & (M1<0.5 | M1>0.6) 
b. Similarly, for the other values of Additive: 

Additive == “Additive 2” & (M1<0.2 | M1>0.6) 
Additive == “Additive 3” & (M1<0.2 | M1>0.5)  
Additive == “Additive 4” & (M1<0.1 | M1>0.3) 
Additive == “Additive 5” & (M1<0.0 | M1>0.2) 
Additive == “Additive 6” & (M1<0.2 | M1>0.4) 
Additive == “Additive 7” & (M1<0.3 | M1>0.5) 
Additive == “Additive 8” & (M1<0.1 | M1>0.4) 

c. Note that the ranges shown above are merely examples, and do not represent actual material ranges. 
 

  



4. Pull the above constraints together into the Disallowed Combinations Script window.  Each constraint is an “OR” 
condition, so the final code looks like this: 

( M1+M2+M3+M4 > 1.01  |  M1+M2+M3+M4 < 0.99 ) | 
( Additive == “Additive 1” & (M1<0.5 | M1>0.6) ) | 
( Additive == “Additive 2” & (M1<0.2 | M1>0.6) ) | 
( Additive == “Additive 3” & (M1<0.2 | M1>0.5) ) | 
( Additive == “Additive 4” & (M1<0.1 | M1>0.3) ) | 
( Additive == “Additive 5” & (M1<0.0 | M1>0.2) ) | 
( Additive == “Additive 6” & (M1<0.2 | M1>0.4) ) | 
( Additive == “Additive 7” & (M1<0.3 | M1>0.5) ) | 
( Additive == “Additive 8” & (M1<0.1 | M1>0.4) ) 
 

5. Create the model.  We wanted to include curvature and interactions in the model.  Since this is a mixture model, 
we chose a Scheffe Cubic model.  (A future blog post will talk about why these terms result from a Scheffe Cubic 
model for mixtures.)  This model includes the following terms: 

a. Five main effects terms (one for each factor) 
 Additive 
 M1 
 M2 
 M3 
 M4 

b. Four two-way interactions for the Additive type vs. each mixture component.  These support : 
 Additive*M1 
 Additive*M2 
 Additive*M3 
 Additive*M4 

c. Six two-way interactions for each pairing of mixture terms  
 M1*M2 
 M1*M3 
 M1*M4 
 M2*M3 
 M2*M4 
 M3*M4 

d. Four 3-way interactions between mixture components: 
 M1*M2*M3 
 M1*M2*M4 
 M1*M3*M4 
 M2*M3*M4 

e. Six terms that handled squared mixture components in an interaction form: 
 M1^2*M2 
 M1^2*M3 
 M1^2*M4 
 M2^2*M3 
 M2^2*M4 
 M3^2*M4 

 



Other Design Parameters 
JMP’s Custom Designer allows input of several more parameters to customize the designer’s behavior.  We chose the 
following: 

Set Random Seed( 0 ) 

This allows the routine to use different “starting points” when finding a design.  Things like run order and even details 
about factor settings can change slightly from design generation to generation.  (Set this to any integer to have JMP 
always reproduce the same design.) 

Number of Starts(100 ) 

JMP uses an iterative solution based on a random starting point to design a test, then starts over with a different 
starting point to see if it can find an improved design.  Increasing this parameter can have the effect of finding a better 
design, at the expense of longer computation times. 

Set Sample Size(65), 

This is the total number of trials in the test.  It must be at least as large as the total number of parameters needed for a 
solution to the model.  Larger number of trials can increase computation time, but more importantly require more 
experimental trials to be made. 

Optimality Criterion( Name( "Make D-Optimal Design" ) ), 

D-Optimal designs give the best estimates of the parameters in the resulting model.  I-Optimal designs give the lowest 
average prediction variance resulting from the model. 

Simulate Responses( 0 ),  

JMP can create simulated responses if you want to test the design.  These have nothing to do with real world results.  
We chose not to simulate responses. 

Save X Matrix( 0 ),  

JMP can save the Moments Matrix and Model Matrix if desired.  We left this option turned OFF. 

 

  



The complete JMP script to handle this design 
As we developed the design, we found it easier to adjust the parameters via editing a script and running the script.  The 
script was first created by creating a manual design, and then using “Save Script to Script Window” from the red triangle.  
The resulting script (after some reformatting) is shown below, and attached to this video: 

DOE( 
 Custom Design, 
 { 
  Add Response( Maximize, "Y", ., ., . ), 
  Add Factor( Categorical, {"Additive 1", "Additive 2", "Additive 3", "Additive 4", 
"Additive 5", "Additive 6", “Additive 7”, “Additive 8” }, "Additive", 0 ), 
  Add Factor( Continuous, 0, 1, "M1", 0 ),  
  Add Factor( Continuous, 0, 1, "M2", 0 ), 
  Add Factor( Continuous, 0, 1, "M3", 0 ),  
  Add Factor( Continuous, 0, 1, "M4", 0 ), 
  
  // Manually enter all main effects 
  Add Term( {1, 1} ), // Additive main effect 
  Add Term( {2, 1} ), // M1 main effect 
  Add Term( {3, 1} ), // M2 main effect 
  Add Term( {4, 1} ), // M3 main effect 
  Add Term( {5, 1} ), // M4 main effect 
  
  // Manually enter all 2-way interactions between mixture components 
  Add Term( {2, 1}, {3, 1} ), // M1*M2 interaction 
  Add Term( {2, 1}, {4, 1} ), // M1*M3 interaction 
  Add Term( {2, 1}, {5, 1} ), // M1*M4 interaction 
  Add Term( {3, 1}, {4, 1} ), // M2*M3 interaction 
  Add Term( {3, 1}, {5, 1} ), // M2*M4 interaction 
  Add Term( {4, 1}, {5, 1} ), // M3*M4 interaction 
  
  // Manually enter all interactions between Additive and mixture components 
  Add Term( {1, 1}, {2, 1} ), // Additive*M1 interaction 
  Add Term( {1, 1}, {3, 1} ), // Additive*M2 interaction 
  Add Term( {1, 1}, {4, 1} ), // Additive*M3 interaction 
  Add Term( {1, 1}, {5, 1} ), // Additive*M4 interaction 
   
  // Manually enter all 3-way interactions 
  Add Term( {2, 1}, {3, 1}, {4, 1} ), // M1*M2*M3 3-way interaction 
  Add Term( {2, 1}, {3, 1}, {5, 1} ), // M1*M2*M4 3-way interaction 
  Add Term( {2, 1}, {4, 1}, {5, 1} ), // M1*M3*M4 3-way interaction 
  Add Term( {3, 1}, {4, 1}, {5, 1} ), // M2*M3*M4 3-way interaction 
   
  // Manually enter appropriate Scheffe Cubic interactions 
  Add Term( {2, 2}, {3, 1} ), // M1^2*M2 interaction 
  Add Term( {2, 2}, {4, 1} ), // M1^2*M3 interaction 
  Add Term( {2, 2}, {5, 1} ), // M1^2*M4 interaction 
  Add Term( {3, 2}, {4, 1} ), // M2^2*M3 interaction 
  Add Term( {3, 2}, {5, 1} ), // M2^2*M4 interaction 
  Add Term( {4, 2}, {5, 1} ), // M3^2*M4 interaction 
 
   
  // Now add the disallowed combinations: 
  Disallowed Combinations 
   ( 
    (M1 + M2 + M3 + M4 > 1.01 | M1 + M2 + M3 + M4 < 0.99) |  
    (Additive == "Additive 1" & (M1 < 0.5 | M1 > 0.6)) |  
    (Additive == "Additive 2" & (M1 < 0.2 | M1 > 0.3)) |  
    (Additive == "Additive 3" & (M1 < 0.2 | M1 > 0.5)) |  
    (Additive == "Additive 4" & (M1 < 0.1 | M1 > 0.3)) |  



    (Additive == "Additive 5" & (M1 < 0.0 | M1 > 0.2)) |  
    (Additive == "Additive 6" & (M1 < 0.2 | M1 > 0.4)) | 
    (Additive == "Additive 7" & (M1 < 0.3 | M1 > 0.5)) | 
    (Additive == "Additive 8" & (M1 < 0.1 | M1 > 0.4)) 
   ),  
 
  Set Random Seed( 0 ),  
  Number of Starts(100 ),   
  Set Sample Size(65), 
  Optimality Criterion( Name( "Make D-Optimal Design" ) ), 
  Simulate Responses( 0 ),  
  Save X Matrix( 0 ),  
  Make Design} 
); 
 

  



Design Diagnostics 
The diagnostics on this design are interesting.  First, let’s look at how well we achieved the constraints of the test on the 
M1 additives: 

 

 

In Figure 1, the blue “whiskers” indicate the desired ranges for each of the M1 additive types, as input in the script.  The 
red dots indicate the actual design points.  As can be seen, the Custom Designer tried to spread the points out across 
each range but didn’t completely cover the desired ranges.  Increasing the Number of Starts can improve the spread of 
the points, but it seemed as though the ranges were never completely covered.  This has been reported to the JMP 
Development team, in hopes that it may be addressed in future software releases. 

  



The color map on correlations produced the following plot: 

 

 

 

Other Observations 
Design Convergence 
Sometimes custom designer didn’t converge.  Increasing NStarts and/or rerunning generally resolved this problem. 

Minimum Trials when running a Script 
Normally JMP’s Custom Designer will provide the minimum number of independent experimental trials required to solve 
for the unknown coefficients in the designated model.  We don’t know a priori what this value is when running the 
script.  Fortunately, if Sample Size is set to some small value (e.g. 2), the script defaults to the minimum number of trials 
required to solve for the coefficients.  So our suggestion is to run the design once to determine the minimum number of 



trials, then increase the Sample Size to some number greater than the min trials (in order to estimate pure error and 
goodness of fit) and rerun the design.  In the case studied in this paper, the minimum number of trials was 56, and the 
chosen Sample Size was 65. 

Savings vs. a Full Factorial Design 
Since we are looking for curvature in this design, a full factorial would have required: 

 8 additive types * 3 M1 levels * 3 M2 * 3 M3 * 3 M4 = 648 trials 

Therefore the Custom Design saved nearly 90% of the trials that would have been required in a full factorial test. 

 
 

Analysis Notes 
 
Prediction Profiler 
Since the input factors were entered as Continuous (as opposed to Mixture) factors, we need to now tell JMP of the 
mixture relationship that all mixture factors must sum to 1.  The easiest way to do this is to change the Design Role 
property for each of the Mixture columns to “Mixture,” and activate the Mixture property and ensure that the sum adds 
up to 1.0. 

Intercept 
When running Fit Model to estimate the model and produce the Prediction Profiler, we need to turn OFF the intercept.  
This is due to the Scheffe Cubic model for mixtures, which does not require an intercept.  (This will be addressed in an 
upcoming JMP Blog post.) 

Limit of Detection 
The measurement of Separation Index in this experiment had a Limit of Detection of 0.001 units.  JMP 16 Pro can handle 
this by setting the “Detection Limit” property for that particular response column, and then using Generalized 
Regression for the analysis.  This gives significantly different results than if Least Squares Regression is used. 

Extrapolation Control 
We want to use the Prediction Profiler to find an optimum setting of Additive type, M1 (Additive Level), M2, M3, and 
M4.  The Prediction Profiler can do this, but we want to restrict the ranges of the additive levels to be appropriate for 
those seen in the test.  In JMP Pro, the Extrapolation Control feature in the Prediction Profiler is intended to cover this 
functionality.  Extrapolation Control can be found under the Prediction Profiler red drop-down triangle menu. 

Unfortunately, it appears that Extrapolation Control does not work with all types of analyses and would not work in our 
case.  This is being reported to the JMP development team. 

 

Testing Notes 
Requirements at Fuchs Lubricants changed as the test was being designed.  In the end, the actual experiment did not 
correspond to the script shown above.  In fact, it was greatly simplified to involve only seven additive types, and no 
other mixture variables.  It was also simplified to reduce the model complexity by only looking at main effects terms (as 
in a screening test). 

This changed the actual test from a mixture design to a simple two-factor design, with one categorical factor (Additive, 
with 7 levels) and different levels for each additive type.  This can be handled in one of two ways: 

 A series of seven independent experiments, one for each Additive type, with two trials each (a high and low 
level).  This results in 14 trials, not including replicates. 



 A combined 14-trial test with two factors (categorical Additive and continuous M1).  The design includes the 
constraints on the levels of M1, but no mixture sum since there is only one component.  The experimental trials 
are the same 14 trials as mentioned above, but the analysis changes somewhat.  Model terms include the two 
main effects (Additive and M1) and the Additive*M1 interaction.  In this case the Intercept term is left On for 
the analysis. 

Repeated trials were also desired in this test, but were not available at the time of this writing.  Therefore, results were 
“exact,” with no uncertainty estimates. 


